
1. [bookmark: _Toc122434485][bookmark: _Toc95247796]Overview
This document lists all the changes needed to correct issues in the ATS iwd-TTCN3-B2020-09_D12wk49 related to the title of this CR.
Contact:	Shaun Harry
	shaun.harry@keysight.com


1. [bookmark: _Toc122434488][bookmark: _Toc295288959][bookmark: _Toc95247797]Corrections required
1. [bookmark: _Toc30685521][bookmark: _Toc83829269][bookmark: _Toc95247798]Change 1
	Function name
	fl_TCP_Close()

	Reason for change
	In some scenarios (including emergency call release) the SS and UE may be attempting to close multiple TCP sockets at the same time. In the TTCN the call to f_TCP_Close() is usually blocking waiting for a confirmation from SS. Any sockets closed by the UE during this time will generate an IP_SOCKET_IND which will then not be handled and will remain in the message queue, blocking the eventually IP_SOCKET_CNF and causing test case execution to hang.


	Summary of change
	Add an option in f_TCP_Close() to absorb any TCP close indication coming while waiting for TCP close cnf, even for a different TCP connection.

Note : This is an extension of the correction proposed under R5s220106. In that CR the main scenario being addressed was the SS and UE closing the same TCP socket at the same time. This CR is addressing the more general case that the UE may be concurrently closing a different TCP socket, but the result is the same and will cause test case execution to hang. So the correction of R5s220106 may no longer be required following the correction in the current CR. The “before” code below is based on R5s220106 implemented.

	TTCN module
	TCP_Functions.ttcn

	MCC160 Comment
	Accepted in principle: In general it is not a proper solution just to skip the TCP_CLOSE_INDs as resources need to be released.
 The following changes are proposed (see below)
a) while waiting for confirmation for TCP_CLOSE_CNF corresponding to the TCP_CLOSE_REQ any other ASPs shall be send back to the IPMUX_PTC where they are queued
b) After the TCP_CLOSE_CNF corresponding to the TCP_CLOSE_REQ has been the received a "flush" command is sent to the IPMUX_PTC which resends the queued ASPs



Before Change
	function fl_TCP_Close(template (value) IP_Connection_Type p_IP_Connection,
boolean p_WaitForCNF) runs on IP_PTC
{
IP_SOCK_CTRL.send(cs_TCP_CLOSE_REQ(p_IP_Connection));
if (p_WaitForCNF) {
alt {
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(p_IP_Connection)) { }
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(?)) { repeat; } /* @sic R5s170043: pending CNF for previous call with p_WaitForCNF==false sic@
* NOTE: in general the CNFs in case of p_WaitForCNF==false may occur at any time */
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_IND(cr_TCP_ConnectionId (p_IP_Connection.Local, *))) { repeat; }
}
}
}    



After Change
	function fl_TCP_Close(template (value) IP_Connection_Type p_IP_Connection,
boolean p_WaitForCNF) runs on IP_PTC
{
IP_SOCK_CTRL.send(cs_TCP_CLOSE_REQ(p_IP_Connection));
if (p_WaitForCNF) {
alt {
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(p_IP_Connection)) { }
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(?)) { repeat; } /* @sic R5s170043: pending CNF for previous call with p_WaitForCNF==false sic@
* NOTE: in general the CNFs in case of p_WaitForCNF==false may occur at any time */
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_IND(cr_TCP_ConnectionId (p_IP_Connection.Local, *))) { repeat; }
[] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_IND(?)) { repeat; }
}
}
}    


MCC160 Implementation
	 function fl_TCP_Close(template (value) IP_Connection_Type p_IP_Connection,
                        boolean p_WaitForCNF) runs on IP_PTC
  {
    var template (present) IP_Connection_Type v_IP_Connection := p_IP_Connection;
    var IP_SOCKET_IND v_IP_SOCKET_IND;

    if (not isvalue(p_IP_Connection.Remote)) {                                   /* stop server; @sic R5s220106: if client tries in parallel to close connection(s) to the server, the remote socket is not omit in any corresponsding close indication sic@ */ 
      v_IP_Connection.Remote := ?;
    }

    IP_SOCK_CTRL.send(cs_TCP_CLOSE_REQ(p_IP_Connection));
    if (p_WaitForCNF) {
      alt {
        [] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(p_IP_Connection)) {
          IP_SOCK_CTRL_QUEUE.send(crs_IP_FLUSH_IND);
        }
        [] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_CNF(?)) { repeat; }                 /* @sic R5s170043: pending CNF for previous call with p_WaitForCNF==false sic@
                                                                                  * NOTE: in general the CNFs in case of p_WaitForCNF==false may occur at any time */
        [] IP_SOCK_CTRL.receive(cr_TCP_CLOSE_IND(v_IP_Connection)) { repeat; }   /* @sic R5s150270 change 1, R5s220106: v_IP_Connection instead of p_IP_Connection sic@ */
        [] IP_SOCK_CTRL.receive(cr_IP_SOCKET_IND_AnyCtrl) -> value v_IP_SOCKET_IND {
          IP_SOCK_CTRL_QUEUE.send(v_IP_SOCKET_IND);
          repeat;
        }
      }
    }
  }



	  type record IP_SOCKET_CTRL_IND {
    IP_Connection_Type          ConnectionId  optional,
    IP_CtrlIndication_Type      Ind  optional
  };

	  template (value) IP_SOCKET_IND crs_IP_FLUSH_IND := 
  {
    CTRL := {
      ConnectionId := omit,
      Ind := omit
    }
  };



	  type component IP_PTC {               /* @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    var IP_PTC_DRB_RoutingInfo_Type     vc_IP_PTC_DRB_RoutingInfo;     /* @sic R5s200490: vc_IP_PTC_DRB_RoutingInfo instead of vc_IP_PTC_DRB_RoutingTable sic@ */
    var PDN_IP_AddrServer_Type          vc_IP_PTC_PDN_AddrServer;
    var IP_PTC_IMS_Info_Type            vc_IP_PTC_IMS_Info;
    var PDN_MappingTable_Type           vc_IP_PTC_PDN_MappingTable;    /* @sic R5s150648 Change 2.1 sic@ */
    var IP_PTC_WLAN_Info_Type           vc_IP_PTC_WLAN_Info;

    // ports connected to IMS:
    port IP_IMS_CTRL_PORT       IMS_CTRL[tsc_NoOfIMSPDNs];             /* @sic R5s201177: tsc_NoOfIMSPDNs instead of tsc_NoOfPDNs sic@ */
    port IP_IMS_SERVER_PORT     IMS_Server[tsc_NoOfIMSPDNs];           /* @sic R5s201177: tsc_NoOfIMSPDNs instead of tsc_NoOfPDNs sic@ */
    port IP_IMS_CLIENT_PORT     IMS_Client[tsc_NoOfIMSPDNs];           /* @sic R5s201177: tsc_NoOfIMSPDNs instead of tsc_NoOfPDNs sic@ */

    // coordination ports:
    port IP_PTC_CTRL_PORT       EUTRA_CTRL;
    port IP_PTC_CTRL_PORT       UTRAN_CTRL;
    port IP_PTC_CTRL_PORT       GERAN_CTRL;
    port IP_PTC_CTRL_PORT       WLAN_CTRL;  /* @sic R5w160107 sic@ */
    port IP_PTC_CTRL_PORT       IPCAN_CTRL; /* @sic R5-203077: MCX IP test model - common IPCAN port independent from test model sic@ */
    port IP_PTC_CTRL_PORT       NR_CTRL;

    // coordination ports to HTTP_PTC
    port IP_HTTP_CTRL_PORT      HTTPCTRL;
    port IP_HTTP_DATA_PORT      HTTPDATA;

    // coordination port for sending of UDP packets to the UE (e.g. for OMA Push)
    port IP_UDP_HANDLER_PORT    UDPHANDLER;

    // coordination port for generic TCP handler
    port IP_TCP_HANDLER_PORT    TCPHANDLER;

    // ports to IP_MUX_PTC
    port IP_SOCKET_PORT         IP_SOCK_DATA;           /* @sic R5s150900 - MCC160 implementation sic@ */
    port IP_SOCKET_PORT         IP_SOCK_CTRL;           /* @sic R5s150900 - MCC160 implementation sic@ */

    port SOCKET_PORT_IPMUX      IP_SOCK_CTRL_QUEUE;

    // system ports
    port IP_CONTROL_PORT        IP_CTRL;
    port IPSEC_CONTROL_PORT     IPSEC_CTRL;
  };

	  type component IPMUX_PTC {                            /* @sic R5s150900 - MCC160 implementation sic@
                                                           @status    APPROVED (ENDC, IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, NR5GC, NR5GC_IRAT, POS, UTRAN) */
    // ports to IP_PTC
    port SOCKET_PORT_IPMUX      IP_SOCK_DATA;
    port SOCKET_PORT_IPMUX      IP_SOCK_CTRL;
    port SOCKET_PORT_IPMUX      IP_SOCK_CTRL_QUEUE;
    // system port
    port IP_SOCKET_PORT         IP_SOCK;
  };

	  type record of IP_SOCKET_IND IP_SOCKET_IND_List;

	  altstep a_IPMUX_PTC_HandlerUL_Queue(inout IP_SOCKET_IND_List p_IP_SOCKET_IND_List) runs on IPMUX_PTC
  {
    var IP_SOCKET_IND v_IP_SOCKET_IND;
    var integer i;

    [] IP_SOCK_CTRL_QUEUE.receive(crs_IP_FLUSH_IND) {
      for (i := 0; i < lengthof(p_IP_SOCKET_IND_List); i:= i + 1) {
        IP_SOCK_CTRL.send(p_IP_SOCKET_IND_List[i]);
      }
      p_IP_SOCKET_IND_List := {};
    }
    [] IP_SOCK_CTRL_QUEUE.receive(cr_IP_SOCKET_IND_AnyCtrl) -> value v_IP_SOCKET_IND {
      i := lengthof(p_IP_SOCKET_IND_List);
      p_IP_SOCKET_IND_List[i] := v_IP_SOCKET_IND;
    }
  }

	  function f_IPMUX_PTC_MainLoop() runs on IPMUX_PTC
  {
    var IP_SOCKET_IND_List v_IP_SOCKET_IND_List_ByRef := {};
    while (true) {
      alt {
        [] a_IPMUX_PTC_HandlerDL(IP_SOCK_DATA);
        [] a_IPMUX_PTC_HandlerDL(IP_SOCK_CTRL);
        [] a_IPMUX_PTC_HandlerUL_Queue(v_IP_SOCKET_IND_List_ByRef);
        [] a_IPMUX_PTC_HandlerUL();
      }
    }
  }

	  function f_IP_PTC_CreateAndMap(IP_SYSTEM p_System) return IP_PTC
  {
    var IP_PTC v_IP_PTC := IP_PTC.create("IP") alive;                // @sic R5s210270 sic@
    var IPMUX_PTC v_IPMUX_PTC := IPMUX_PTC.create("IPMUX") alive;    // @sic R5s210270 sic@

    map(v_IP_PTC:IP_CTRL,    p_System:IP_CTRL);
    map(v_IP_PTC:IPSEC_CTRL, p_System:IPSEC_CTRL);
    map(v_IPMUX_PTC:IP_SOCK, p_System:IP_SOCK);

    connect(v_IP_PTC:IP_SOCK_DATA, v_IPMUX_PTC:IP_SOCK_DATA);
    connect(v_IP_PTC:IP_SOCK_CTRL, v_IPMUX_PTC:IP_SOCK_CTRL);
    connect(v_IP_PTC:IP_SOCK_CTRL_QUEUE, v_IPMUX_PTC:IP_SOCK_CTRL_QUEUE);

    v_IPMUX_PTC.start(f_IPMUX_PTC_MainLoop());

    return v_IP_PTC;
  }





